NASA TECHNICAL MEMORANDUM

NASA TM X-64742

THE GAERTNER L119 ELLIPSOMETER AND ITS USE IN THE MEASUREMENT OF THIN FILMS

By Michael Linkous
Space Sciences Laboratory

March 16, 1973

NASA

George C. Marshall Space Flight Center
Marshall Space Flight Center, Alabama

TECHNICAL REPORT STANDARD TITLE PAGE

1. REPORT NO.	2. GOVERNMENT ACCESSION NO.	3. RECIPIENT'S CATALOG NO.
NASA TM X-64742		

4. TITLE AND SUBTITLE The Gaertner L119 Ellipsometer and Its Use in the Measurement of Thin Films	5. REPORT DATE March 16, 1973 6. PERFORMING ORGANIZATION CODE
7. AUTHOR(S) Michael Linkous	8. PERFORMING ORGANIZATION REPORT #
9. PERFORMING ORGANIZATION NAME AND ADDRESS George C. Marshall Space Flight Center Marshall Space Flight Center, Alabama 35812	10. WORK UNIT NO. 11. CONTRACT OR GRANT NO.
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration Washington, D. C. 20546	13. TYPE OF REPORT & PERIOD COVERED Technical Memorandum 14. SPONSORING AGENCY CODE

15. SUPPLEMENTARY NOTES

Prepared by Space Sciences Laboratory,
Science and Engineering

16. ABSTRACT

An introduction to the study of ellipsometry is presented, with special attention given to the Gaertner model L119 ellipsometer and the techniques of measuring thin films with this instrument.

Values obtained from the ellipsometer are analyzed by a computer program for a determination of optical constants and thickness of the film.

17. KEY WORDS	18. DISTRIBUTION STATEMENT
	Unclassified - unlimited Michael A. Linkous

19. SECURITY CLASSIF. (of this report)	20. SECURITY CLASSIF. (of this page)	21. NO. OF PAGES	22. PRICE
Unclassified	Unclassified	43	NTIS

MSFC - Form 3292 (Rev December 1972) For sale by National Technical Information Service, Springfield, Virginia 22151

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

TECHNICAL MEMORANDUM X-64742

THE GAERTNER L119 ELLIPSOMETER AND ITS USE IN THE MEASUREMENT OF THIN FILMS

DESCRIPTION

The optical parameters for a thin film present on a reflecting surface may be ascertained by ellipsometry [1]. Using these values, a computer program based on iterative solutions to the Fresnel equations can calculate the index of refraction, the coefficient of absorption, and the thickness of the film if the index of the reflecting surface and of the ambient medium is known [2]. The intention of this paper is to discuss ellipsometry and the model L119 Gaertner ellipsometer facility (Fig. 1) which is available in the Thermal Environment Physics Branch of the Space Sciences Laboratory at Marshall Space Flight Center. Interest in this facility has made apparent the desirability of such an explanation.

Ellipsometry deals with the change in the state of the polarization of reflected light [3]. The ellipsometer measures this change by the orientation of two opposing Glan-Thompson prisms [4], as shown in Figure 2.

Incident light is plane polarized by an initial Glan-Thompson prism, the polarizer. Its phase is then changed 90 degrees by a $1/4\lambda$ plate and reflected by the surface under investigation to a second Glan-Thompson prism which analyzes the ellipticity of the reflected light. A photomultiplier increases the sensitivity of the vernier readings for the P and A values [3].

The light source is a mercury lamp that emits unpolarized light unsuitable for use with the ellipsometer. An interference filter between the light source and the polarizer passes a narrow band [546.1 nm (5461 Å)] of light which is compatible with the compensator. The light is collimated by a series of baffles before encountering the polarizer. The polarizer is able to select a single plane of vibration because its index and coefficient of absorption is different for light polarized in different directions [5].



Figure 1. Gaertner ellipsometer L119.



Figure 2. Schematic of the L119 ellipsometer.

A mica 1/4-wave plate standardizes the degree of the polarization of the reflected light by changing the phase of the incident light by 90 degrees, or quarter wavelength [4]. Because the crystallographic and optical axes of mica are complementary, light entering the compensator, except parallel to the axes, will be broken into two new components vibrating parallel to the axes, as shown in Figure 3 [6]. These components will travel at different speeds. One (the ordinary ray) will continue to travel the same distance in the same time as ordinary light. The other component (the extraordinary ray) will travel in the same time a different distance [6]. In this case the distance is the thickness of the mica plate, which for a wavelength of 546.1 nm (5461 Å) changes the phase of the light by 90 degrees [1].

A change in phase of 90 degrees constitutes circularly polarized light. This is the state of polarization of the light incident to the surface in question.

Upon reflection, the phase of the light will be further changed because of the reflecting material's index of refraction and its coefficient of absorption [5]. The incident light is again broken into two components, one of which is reflected directly from the surface, while the other is absorbed by the material and then reflected, as shown in Figure 4. As with the compensator, one component of light has traveled farther than the other [3].



Figure 3. Path of light through mica 1/4 λ plate.



Figure 4. Absorption and reflection of polarized light from a film-covered substrate [3].

When the light is refocused, its phase will again have been changed; however, the incident light is circularly polarized to compensate for this change in phase by reflection. The electric vectors of the resulting reflected light at any given point along the propagation of the wave will appear to trace an ellipse [7]. The light can now be analyzed by a second Glan-Thompson prism. This represents the A value. The acuteness of the ellipticity of the light has been compensated for, and the resulting wave is well within the limit of measurement by the analyzer.

After being analyzed, the light is focused into a photomultiplier, enabling the operator to mechanically determine the extinction coefficients of the reflecting surface [4]. Once the L119 is aligned, the technique for obtaining P and A values is simple. The sample must be mounted securely, insuring proper alignment. Any correction for alignment should be done by the three leveling screws beneath the sample table. The angle of incidence may be set by rotating the telescope until twice the desired angle is read on the vernier beneath the sample table. A Gauss eyepiece inserted in place of the photomultiplier may be used to align and focus the light [3]. Although a darkened room is not required, it is best that a minimum of room lighting be used. Care should be taken not to expose the photomultiplier to intense light since this will adversely affect it and result in erratic readings.

To obtain a P value, the polarizer vernier should be slowly rotated until the picoammeter shows a minimum or maximum of intensity. This setting may be more accurately read by taking an average of settings that give equal intensities on either side of the minimum [3]. A reading of the analyzer may be taken in the same way. Only those values in zones 1 (0 to 90 degrees) and 2 (90 to 180 degrees) need to be calculated for use by the computer program. These values must fall between 0 and 180 degrees, and are designated P_1 and P_2 for the polarizer and A_1 and A_2 for the analyzer. It can be determined that $P_2 \approx P_1 + 90$ degrees and that $A_1 + A_2 \approx 180$ degrees [3]. Whether a value appears as a maximum or minimum is of little consequence since any value recorded by one vernier as a maximum may be read again as a minimum, depending on the orientation of the opposing vernier. The accuracy of the L119 may be checked by setting the angle between the telescope and collimator at 180 degrees. If the polarizer is set at 0 or 180 degrees, extinction of the light should occur at 90 or 270 degrees on the analyzer. Some discrepancy has been noted between these ideal extinction settings and those actually occurring with the facility at MSFC. In some instances there are discrepancies of as much as 30 degrees. $A_1 + A_2$ usually equals from 200 to 215 degrees. This may be caused by improper alignment or any number of varying factors.

Table 1 consists of data taken from a stainless steel mirror and shows the repeatability of the instrument. As shown, the L119 is repeatable within 1 degree. However, repeatability to the nearest 0.01 degree is in the range of this instrument. Again, discrepancies may be caused by misalignment or, more probably, the accuracy with which the operator makes the readings. Nevertheless, the accuracy of the instrument is severely hindered by the unsteadiness of both the collimator and telescope arms. Even after they are locked into position, the arms of the telescope may slip as much as 0.635 centimeters during a reading. As shown in Figure 5, a 10-degree change in the polarizer reading increased the thickness of that particular mirror by almost 8.0 nanometers and a similar change in the analyzer resulted in a difference of 2.0 nanometers.

These values were computed by an 1108 computer using a Fortran program [2]. A change in the polarizer results in a greater change in the computed values than does a similar change in the analyzer. The reason for this is shown in Figure 6. First, note the difference in the shape of the curves for the analyzer and polarizer for the two zones shown. The slope of the A curve is much smoother than the P curve. Changes in A occur over a wider range than do similar changes in P. Also note that a value at $P_1 + 90$ degrees is approximately equal to the P_2 value at the same angle. However, the $A_1 + A_2$ value for a given angle in every case greatly exceeds 180 degrees.

Included as Appendix A is a computer output for a program revised from the program mentioned above to run on the 1108 computer facility at MSFC.

TABLE 1. P AND A VALUES FOR ONE ZONE SHOWING REPEATABILITY FOR CLEAN STAINLESS STEEL MIRROR

P	A
244.45	16.60
244.30	16.74
244.45	16.60
244.45	16.69
244.42	16.63
244.40	16.69
244.40	16.77
244.30	16.81
244.40	16.82
244.46	16.35
244.55	16.59
244.42	16.57
244.54	16.59
244.56	16.63
244.60	16.59
244.54	16.65
244.55	16.57
244.50	16.60
244.57	16.57
244.60	16.63
244.71	16.70
244.60	16.70
244.65	16.72
244.80	16.68
244.63	16.55
244.43	16.64
244.30	16.59

Appendix B is a graph showing the same type data as shown in Figure 6. It compares a mono-layered aluminum mirror with a multilayered lithium-fluoride on aluminum mirror as a function of their optical parameters and the angle of the incident light.

Shown in Appendix C are the results of the parameters of several mirrors measured as functions of the $1/4\lambda$ plate and as a function of each other. The calculation of small delta (δ) as used in the fresnel equations for a film on silicon is shown as a function of thickness in Appendix D. Shown in Appendix E is a graph of the same data displaying an elliptical pattern.



Figure 5. Polarizer and analyzer headings as a function of the thickness of an imaginary film.



Figure 6. P and A values versus angle of incidence for aluminum mirror.

APPENDIX A

FORTRAN IV PROGRAM

@HDG, P
@FOR, IS ELLIPS

```
C       CALCULATION FOR ELLIPSOMETER
C         FOR IFILM, THE NUMBER OF FILMS ADD. RESET IS NECESSARY TO ERASE FILMS
C     CHANGED TO USE KAPPA INSTEAD OF IMAGINARY PART OF REFRACTIVE INDEX
      DIMENSION IDT(14),FNSI(2),DFNS(2),FNSE(2),FNSX(2),TN(8),TD(8)
      DIMENSION INST(31)
      COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
     1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
     2 T4,D2,DELDX,PSIDX, ELDEL,ELPSI,DENP,FNP,DEN1,MX,K,IS
      COMPLEX C2,C3,CI,CIP1,CN,CNM1,CRN,CRN21,CRN32,CRP,CRP21,CRP32
      COMPLEX FN2,FNX,FNS,  RHO,RN,RN21,      RP,RP21,      RC,SRC
      COMPLEX T1,T2,T3,T4,RHOC,RHO1,RHO2,CAT,ALP1,ALP2,BET1,BET2
      COMPLEX GAM1,GAM2,CAT
C             INSTRUCTIONS TO BE COMPARED WITH INSTRUCTIONS GIVEN WITH
C                                     DATA
C            M=        1        2        3        4        5        6
      DATA INST/6HPAGE   ,6HNM    ,6HNF    ,6HNS    ,6HAI    ,6HWL    ,
C           7        8        9        10       11       12       13
     1 6HWP    ,6HNREF  ,6HEDP   ,6HELDP  ,6HRESET ,6HFILM  ,6HMIXED ,
C          14       15       16       17       18       19       20
     2 6HCTABLE,6HCD    ,6HCND   ,6HCNDE  ,6HCNDC  ,6HCNS   ,6HCWP   ,
C          21       22       23       24       25       26       27
     3 6HCDN   ,6HCDNE  ,6HCDNC  ,6HAT    ,6HCAT   ,6HCATA  ,6HIFILM ,
C            28       29       30  31
     4  6HALINE ,6HSTOP  ,2HFT,3HCNK/
C         SRC IS SMALL REFLECTION COEFFICIENT (WITHOUT CONSIDERING
C                 UNDERLYING LAYERS)
      SRC(A,B)=(A-B)/(A+B)
C             SIN,COS,ATAN,TAN FOR ANGLES IN DEGREES
      SIND (X)=SIN(X/57.29578)
      COSD (X)=COS(X/57.29578)
      ATAND (X)=ATAN(X)*57.29578
      TAND(X)=SIND(X)/COSD(X)
C         S IS USED FOR CALCULATING SPECIFIC REFRACTION
      S(XN)=(XN*XN-1.)/(XN*XN+2.)
C             GIVE STANDARD VALUES TO VARIABLES
   10 FN1I=1.
      DFN1=0.
      FN1E=0.
      AT=0.
      ALP=0.
      ALA=0.
      ALQ=0.
      FN2I(1)=1.
      FN2I(2)=0.
      DFN2(1)=0.
      DFN2(2)=0.
      FN2E(1)=0.
      FN2E(2)=0.
      FNSI(1)=1.
      FNSI(2)=0.
      DFNS(1)=0.
      DFNS(2)=0.
      FNSE(1)=0.
      FNSE(2)=0.
      AII=70.
      AIE=0.
      WLI=5461.
```

```
      DWL=0.
      WLE=0.
      RHOC=(0.,-1.)
      TC=1.
      DELC=90.
      DEN1=1.
      DENP=1.
      FNP=1.5
      N=2
      ELDEL=.002
      ELPSI=.002
      D2I=0.
      DD2=0.
      D2F=0.
      FN=1.
      EDEL=.015
      EPSI=.015
C                READ AND PRINT TITLE CARD
   12 READ(5,15) IDT
   15 FORMAT(13A6,A2)
      WRITE(6,17) IDT
   17 FORMAT(1H1,14A6)
C               READ AND PRINT CARD OF DATA
   20 READ(5,25) L,Z1,Z2,Z3,Z4,Z5,Z6
   25 FORMAT(A6,3X,6F10.0)
      WRITE(6,27) L,Z1,Z2,Z3,Z4,Z5,Z6
   27 FORMAT(1H ,A6,3X,6F15.5)
C               COMPARE L WITH INSTRUCTIONS TO FIND M.
      DO 30 M=1,31
   30   IF(L.EQ.INST(M))  GO TO 35
      GO TO 20
C       M=   1  2  3  4  5   6   7   8   9  10 11  12  13  14  15  16
   35 GO TO(12,50,70,80,90,100,110,120,130,140,10,150,160,305,170,170,
C       17  18  19  20  21  22  23  24  25  26  27  28  29  30  31
     1 170,170,170,730,170,170,170,250,750,750,270,260,810,305,170),M
C             INSTRUCTION NM,REFRACTIVE INDEX OF MEDIUM
   50 FN1I=Z1
      DFN1=Z2
      FN1E=1.000001*Z3
      GO TO 20
C         INSTRUCTION NF, REFRACTIVE INDEX OF TOP FILM
   70 FN2I(1)=Z1
      DFN2(1)=Z2
      FN2E(1)=1.000001*Z3
      FN2I(2)=Z4
      DFN2(2)=Z5
      FN2E(2)=1.000001*Z6
C            IF DFN2(1) IS LEFT BLANK ON DATA CARD, SUBSTITUTE A VALUE OF
C                  DFN2(1) SO THAT ONLY ONE CALCULATION WILL BE PREFORMED.
      IF(ABS(DFN2(1)).EQ.0.) DFN2(1)=1.000002*ABS(FN2E(1)-FN2I(1))
C             INSTRUCTION NS, REFRACTIVE INDEX OF SUBSTRATE
   80 FNSI(1)=Z1
      DFNS(1)=Z2
      FNSE(1)=1.000001*Z3
      FNSI(2)=Z4
      DFNS(2)=Z5
      FNSE(2)=1.000001*Z6
      GO TO 20
C             INSTRUCTION AI, ANGLE OF INCIDENCE
   90 AII=Z1
      DAI=Z2
      AIE=1.000001*Z3
```

```
      GO TO 20
C       INSTRUCTION WL,   WAVE LENGTH OF LIGHT
  100 WLI=Z1
      DWL=Z2
      WLE=1.000001*Z3
      GO TO 20
C     INSTRUCTION WP, CONSTANTS OF WAVE PLATE
  110 TC=Z1
      DELC=Z2
      RHOC=TC*CEXP((0.,-1.)*DELC/57.29578)
      GO TO 20
C        INSTRUCTION NREF, NUMBER OF REFLECTIONS OF THE LIGHT
  120 FN=Z1
      GO TO 20
C       INSTRUCTION EDP, CONFIDENCE FIMITS FOR DELTA AND PSI
  130 EDEL=Z1
      EPSI=Z2
      GO TO 20
C      INSTRUCTION ELDP, ERROR LIMITS FOR DELTA AND PSI
  140 ELDEL=Z1
      ELPSI=Z2
      GO TO 20
C    INSTRUCTION FILM ,NUMBER, THICKNESS AND REFRACTIVE INDEX OF FILM
  150 I=1.000001*Z1+1
      D(I)=Z2
      FNX(I)=CMPLX(Z3,-Z3*ABS(Z4))
      N=MAX0(N,I)
      GO TO 20
C      INSTRUCTION MIXED, DATA FOR MIXED FILMS
  160 DEN1=Z1
      DENP=Z2
      FNP=Z3
      GO TO 20
C    VALUES OF DELTA AND PSI OR OF P,A,AND Q HAVE BEEN GIVEN
C        USE PREVIOUS VALUES OF DEL AND PSI IF NO DATA IS GIVEN,Z1=Z2=0
C             IF FIRST VALUE(Z1) =0,USE PREVIOUSLY GIVEN VALUES OF
C                              DELTA AND PSI
  170 IF(ABS(Z1).EQ.0.) GO TO 310
      IF(ABS(Z5).EQ.0.) GO TO 180
C             Z5 NOT =0, SO P1,A1,P2,A2,Q ARE GIVEN
C             COMPUTE TWO VALUES OF RHO
      Z1=Z1+AT+ALP
      Z2=Z2+AT+ALA
      Z3=Z3+AT+ALP
      Z4=Z4+AT+ALA
      Z5=Z5+AT+ALQ
      WRITE(6,27) L,Z1,Z2,Z3,Z4,Z5
  174 RHO1=TAND(Z2)*(TAND(Z5)+TAND(Z1-Z5)*RHOC)/(TAND(Z1-Z5)*TAND(Z5)*
     1     RHOC-1.)
      RHO2=TAND(Z4)*(TAND(Z5)+TAND(Z3-Z5)*RHOC)/(TAND(Z3-Z5)*TAND(Z5)*
     1     RHOC-1.)
EC            AVERAGE VALUES OF RHO,COMPUTE VALUES OF DELTA AND PSI
      RHO=(RHO1+RHO2)/2.
      DEL1=ATAN2(AIMAG(RHO1),REAL(RHO1))*57.29578
      PSI1=ATAND(CABS(RHO1))
      DEL2=ATAN2(AIMAG(RHO2),REAL(RHO2))*57.29578
      PSI2=ATAND(CABS(RHO2))
      WRITE(6,175) DEL1,PSI1,DEL2,PSI2
  175 FORMAT(6H DEL1=,F12.6,6H PSI1=,F12.6,6H DEL2=,F12.6,
     1 6H PSI2=,F12.6)
      IF(FN.EQ.1.) GO TO 210
      GO TO 200
```

```
C              IF Z3=0, DELTA AND PSI ARE GIVEN,OTHERWISE P,A,Q ARE GIVEN
  180 IF(ABS(Z3).EQ.0.) GO TO 190
      Z1=Z1+AT+ALP
      Z2=Z2+AT+ALA
      Z3=Z3+AT+ALQ
      WRITE(6,27) L,Z1,Z2,Z3
  185 RHO=TAND(Z2)*(TAND(Z3)+TAND(Z1-Z3)*RHOC)/(TAND(Z1-Z3)*TAND(Z3)*
     1     RHOC-1.)
      IF(FN.EQ.1.) GO TO 210
      GO TO 200
  190 DEL=Z1
      PSI=Z2
      RHO=TAND(PSI)*CEXP((0.,1.)*DEL/57.29578)
      IF(FN.EQ.1.) GO TO 310
C              CORRECT FOR NUMBER OF REFLECTIONS IF FN NOT =1
  200 RHO=CEXP(1./FN*CLOG(RHO))
  210 DEL=ATAN2(AIMAG(RHO),REAL(RHO))*57.29578
      PSI=ATAND(CABS(RHO))
      IF(M.EQ.25 .OR. M.EQ.26)  GO TO 755
  215 WRITE(6,220) DEL,PSI
  220 FORMAT(6H  DEL=,F13.6,5H PSI=,F13.6)
      GO TO 310
C            INSTRUCTION AT, ANGLE OF TILT OF SAMPLE
  250 AT=Z1
      GO TO 20
C        INSTRUCTION ALINE, ALINEMENT CORRECTIONS
  260 ALP=Z1
      ALA=Z2
      ALQ=Z3
      GO TO 20
C           INSTRUCTION IFILM, CALCULATES THICKNESSES AND REFRACTIVE
C             INDICES OF A SERIES OF FILMS
  270 XNF=Z1
      TH=Z2
C         REMOVE ALL PREVIOULSY DESIGNATED FILMS.
C     WITHOUT NEXT CARD, FILMS WILL BE PLACED BETWEEN PREVIOUSLY
C          DESIGNATED FILMS AND SUBSTRATE
      N=2
C         DN AND DK ARE INCREMENTS OF REFRACTIVE INDEX AND K
      DN=(Z5-Z3)/XNF
      DK=(Z6-Z4)/XNF
C              COMPUTE INDEX AND K (KAPPA) OF TOP FILM
      Z3=Z3+DN/2.
      Z4=Z4+DK/2.
C         INDEX OF TOP FILM IS N1, INDEX OF BOTTOM FILM IS N
      N1=N+1
      NF=1.000001*XNF
      N=N+NF
      DO 280 I=N1,N
      D(I)=TH/NF
      FNX(I)=CMPLX(Z3,-Z3*Z4)
      Z3=Z3+DN
  280 Z4=Z4+DK
      GO TO 20
C     THICKNESS OF TOP FILM AND ,OPTIONALLLY, SETTING OF WAVE PLATE
  305 D2I=Z1
      DD2=Z2
      D2E=1.000001*Z3
      Q=Z4
      IF(M.EQ.30) GO TO 20
C       SET INITIAL  VALUES
  310 FN1=FN1I
```

```
  311 AI=AII
  312 S1=SIND(AI)
      C1=COSD(AI)
      IF(M.EQ.19) GO TO 710
  315  WL=WLI
      WRITE (6,316) FN1,AI,WL,TC,DELC
  316 FORMAT(17H0INDEX OF MEDIUM=,F10.5,21H  ANGLE OF INCIDENCE=,F10.3,
     1 14H  WAVE LENGTH=,F10.2,5H  TC=,F10.5,7H  DELC=,F10.3)
      IF(M.EQ.14) GO TO 320
      IF(M.EQ.31) GO TO 318
C             PRINT TABLE HEADINGS EXCEPT FOR CTABLE OR CNK
      WRITE(6,317)
  317 FORMAT(123H0    THICK  NF REAL    KAPPA  ADSORPTION    D DEL    D P
     1SI       ERROR     NF-NM      NS-NF WT FRACTION  NS REAL   KAPPA
     2 K)
      GO TO 320
C          PRINT HEADINGS FOR INSTRUCTION CNK
  318 PRINT 319
  319 FORMAT(1H0,8X,2HNF,15X,5HKAPPA,9X,4HD NF,11X,7HD KAPPA,9X,
     1  5HERROR,3X,7HD DELTA,5X,5HD PSI,10X,5HTHICK)
  320  NP1=N+1
C             SET VALUES FOR REFRACTIVE INDEX OF SUBSTRATE
  325  FNSX(1)=FNSI(1)
  326  FNSX(2)=FNSI(2)
  327  FNX(NP1)=CMPLX(FNSX(1),-FNSX(2)*FNSX(1))
C         COMPUTE  REFLECTION COEF  BETWEEN LAYERS 3 AND 4
       IF(N.EQ.2) GO TO 335
       X=SIND(AI)*FN1
      CN=CSQRT(1.-(X/FNX(N+1))**2)
      CNM1=CSQRT(1.-(X/FNX(N))**2)
       CRP=SRC(FNX(N+1)*CNM1,FNX(N)*CN)
       CRN=SRC(FNX(N)*CNM1,FNX(N+1)*CN)
       IF(N.EQ.3) GO TO 335
       NX=N-3
       I=N-1
      DO 330 J=1,NX
        CIP1=CSQRT(1.-(X/FNX(I+1))**2)
        CI=  CSQRT(1.-(X/FNX(I))   **2)
        RP=SRC(FNX(I+1)*CI,FNX(I)*CIP1)
        RN=SRC(FNX(I)*CI,FNX(I+1)*CIP1)
        T1=FNX(I+1)*CIP1*D(I+1)/WL
        CRP=RC(RP,CRP,T1)
        CRN=RC(RN,CRN,T1)
  330 I=I-1
  335 IF(M.EQ.31) GO TO 850
  340 FN2X(2)=FN2I(2)
  342 FN2X(1)=FN2I(1)
  343 FN2=CMPLX(FN2X(1),-FN2X(2)*FN2X(1))
C        CALCULATE REFLECTION COEF BETWEEN LAYERS 2 AND 3
      CALL  CRC
      IF(M.EQ.14) GO TO 350
      K=1
C             EXCEPT FOR CTABLE,CALCULATE VALUES,T1 AND T2, OF FILM
C                               THICKNESS
      CALL  CALDX (T1,T2)
      IF(M.EQ.15) GO TO 455
C             EXCEPT FOR CTABLE OR CD,CALCULATE COMPLEX VALUES T3 AND
C                    T4 OF FILM THICKNESS FOR UPPER VALUE OF FILM
C                   REFRACTIVE INDEX
      FN2=CMPLX(FN2X(1)+DFN2(1),-FN2X(2)*(FN2X(1)+DFN2(1)))
      CALL  CRC
      CALL  CALDX(T3,T4)
```

```
C    IS=1 FOR NO SOLUTION FOUND   IS=2 FOR SOLUTION FOUND
      IS=1
      IF(( AIMAG(T1).GT.0..AND.AIMAG(T3).LT.0.).OR.(AIMAG(T1).LT.0..AND.
     1  AIMAG(T3).GT.0.)) GO TO  485
C          NO SOLUTION FOR N2 FOR FIRST SOLUTION OF QUADRATIC
  345 IF((AIMAG(T2).GT.0..AND.AIMAG(T4).LT.0.).OR.(AIMAG(T2).LT.0.AND.
     1  AIMAG(T4).GT.0.))   GO TO 487
C          NO SOLUTION FOR N2 FOR SECOND SOLUTION OF QUADRATIC
  347 IF(IS.EQ.2) GO TO 670
C    NO SOLUTION FOUND, TRY OTHER QUADRANTS FOR SQRT IN CALDX
      K=3
      FN2=CMPLX(FN2X(1),-FN2X(2)*FN2X(1))
      CALL  CRC
      CALL  CALDX (T1,T2)
      FN2=CMPLX(FN2X(1)+DFN2(1),-FN2X(2)*(FN2X(1)+DFN2(1)))
      CALL  CRC
      CALL  CALDX(T3,T4)
      IF(( AIMAG(T1).GT.0..AND.AIMAG(T3).LT.0.).OR.(AIMAG(T1).LT.0..AND.
     1  AIMAG(T3).GT.0.)) GO TO  490
  348 IF((AIMAG(T2).GT.0..AND.AIMAG(T4).LT.0.).OR.(AIMAG(T2).LT.0.AND.
     1  AIMAG(T4).GT.0.))   GO TO 489
      GO TO 670
C         TABLES OF DEL AND PSI FOR CTABLE
  350 WRITE(6,360)
  360 FORMAT(1H0,3X,5HTHICK,4X,3HDEL,5X,3HPSI,4X,14HRF.CF.PARALLEL,6X,
     1         12HRF.CF.NORMAL,4X,2HNF,4X,5HKAPPA,2X,2HNS,5X,5HKAPPA,4H,
     2         2HP1,5X,2HA1,5X,2HP2,5X,2HA2)
  390 D2=D2I
  400 CRP21=RC(RP21,CRP32,FN2*C2*D2/WL)
      CRN21=RC(RN21,CRN32,FN2*C2*D2/WL)
      RHO=CRP21/CRN21
      IF(FN.NE.1.) RHO=CEXP(1./FN*CLOG(RHO))
      DEL=ATAN2(AIMAG(RHO),REAL(RHO))*57.29578
      IF(REAL(RHO).LT.0. .AND. ABS(AIMAG(RHO)).LT. 1.0E-6) DEL=180.
      PSI=ATAND(CABS(RHO))
      XM1=CABS(CRP21)
      THETA1=ATAN2(AIMAG(CRP21),REAL(CRP21))*57.29578
      IF(REAL(CRP21).LT.0. .AND. ABS(AIMAG(CRP21)).LT.1.0E-6)THETA1=180.
      XM2=CABS(CRN21)
      THETA2=ATAN2(AIMAG(CRN21),REAL(CRN21))*57.29578
      IF(REAL(CRN21).LT.0. .AND. ABS(AIMAG(CRN21)).LT.1.0E-6)THETA2=180.
      FN2R=REAL(FN2)
      FNXR=REAL(FNX(N+1))
      FK2=-AIMAG(FN2)/REAL(FN2)
      FKX=-AIMAG(FNX(N+1))/REAL(FNX(N+1))
      IF(ABS(Q).NE.0.) GO TO 420
      WRITE(6,410) D2,DEL,PSI,XM1,THETA1,XM2,THETA2,FN2R,FK2,FNXR,FKX
  410 FORMAT(1H F8.1,2F8.3,2(F9.5,F9.3),4F7.4,4F7.2)
      GO TO 450
C            Q NOT 0, COMPUTE P AND A READINGS FOR VALUE OF Q
C     CORRECT Q FOR TILT AND ALINEMENT
  420 TQ=TAND(Q+AT+ALQ)
      IF(ABS(DEL).LT.1.0E-6) GO TO 430
      R=(TQ*SIND(DEL-DELC)-SIND(DEL+DELC)/TQ)/2./SIND(DEL)
      TP1=(-R+SQRT(R*R+1))/TC
      TP2=(-R-SQRT(R*R+1))/TC
      P1=ATAND(TP1)+Q-ALP+ALQ
      P2=ATAND(TP2)+Q-ALP+ALQ
      A1=ATAND(REAL(RHO*(TP1*TQ*RHOC-1.)/(TQ+TP1*RHOC)))-AT-ALA
      A2=ATAND(REAL(RHO*(TP2*TQ*RHOC-1.)/(TQ+TP2*RHOC)))-AT-ALA
      GO TO 440
C   USE LIMITING FORMS OF EQUATIONS FOR DEL NEARLY =0.
```

```
  430 P1=Q-ALP+ALQ
      P2=Q-ALP+ALQ-90.
      A1=ATAND(REAL(RHO/TQ))-AT-ALA
      A2=ATAND(REAL(RHO*TQ))-AT-ALA
C        REDUCE ANGLES TO LIE BETWEEN 0 AND 180 DEGREES
  440 P1=AMOD(P1+360.,180.)
      A1=AMOD(A1+360.,180.)
      P2=AMOD(P2+360.,180.)
      A2=AMOD(A2+360.,180.)
      WRITE(6,410) D2,DEL,PSI,XM1,THETA1,XM2,THETA2,FN2R,FK2,FNXR,
     1 FKX,P1,A1,P2,A2
  450 D2=D2+DD2
      IF(D2.LT.D2E) GO TO 400
      GO TO  670
C      COMPUTE FILM THICKNESS  M=15        CD
C      PICK THICKNESS WITH  SMALLEST IMAGINARY PART
  455  IF(ABS(AIMAG(T2)).LT.ABS(AIMAG(T1)))   GO TO 460
       D2=REAL(T1)
       ERROR=AIMAG(T1)
      K=1
       GO TO 470
  460  D2=REAL(T2)
       ERROR=AIMAG(T2)
      K=2
C     COMPUTE DEL AND PSI FOR CALCULATED FILM THICKNESS D2
  470  CRP21=RC(RP21,CRP32,FN2*C2*D2/WL)
       CRN21=RC(RN21,CRN32,FN2*C2*D2/WL)
      RHO1=CRP21/CRN21
      PSIX=ATAND(CABS(RHO1))-PSI
      DELX=57.29578*ATAN2(AIMAG(RHO1),REAL(RHO1))-DEL
      DIFN21=REAL(FN2)-FN1
      DIFN32=REAL(FNX(N+1))-REAL(FN2)
C      WEIGHT FRACTION AND WEIGHT OF MATERIAL IN UNIT AREA OF FILM
      FRACTN=DENP/(DENP+(S(REAL(FN2))-S(FNP))*DEN1/(S(FN1)-S(REAL(FN2)))
     1 )
      WTMT=1.0E-5*D2*FRACTN/(FRACTN/DENP+(1.-FRACTN)/DEN1)
      FN2R=REAL(FN2)
      FNXR=REAL(FNX(N+1))
      FK2=-AIMAG(FN2)/REAL(FN2)
      FKX=-AIMAG(FNX(N+1))/REAL(FNX(N+1))
      WRITE(6,480) D2,FN2R,FK2,WTMT,DELX,PSIX,ERROR,DIFN21,DIFN32,
     1 FRACTN,FNXR,FKX,K
  480 FORMAT(1H ,F8.1,2F9.5,1PE12.4,0P2F9.3,F12.5,3F10.5,2F9.5,I6)
      GO TO   670
C        CALCULATE SOLUTION FOR N2 FOR FIRST SOLUTION OF QUADRATIC
  485 K=1
      GO TO 490
  487 K=2
      GO TO 490
  489 K=4
C           MX=16 SO THAT SOLN2 WILL PRINT
  490 MX=16
      CALL SOLN2
      MX=M
      IF(M.EQ.16 .OR. M.EQ.21 .OR. IS.EQ.1) GO TO (345,347,348,670),K
C          PRINT BLANK SPACE
      WRITE(6,27)
C        CALCULATE ERROR LIMITS FOR N2 AND D2
      DEL0=DEL
      PSI0=PSI
      DEL=DEL0+1.249*EDEL
      CALL SOLN2
```

```
      TN(1)=REAL(FN2)
      TD(1)=D2
      DEL=DEL0-1.249*EDEL
      CALL SOLN2
      TN(2)=REAL(FN2)
      TD(2)=D2
      DEL=DEL0
      PSI=PSI0+1.249*EPSI
      CALL SOLN2
      TN(3)=REAL(FN2)
      TD(3)=D2
      PSI=PSI0-1.249*EPSI
      CALL SOLN2
      TN(4)=REAL(FN2)
      TD(4)=D2
      DEL=DEL0+0.883*EDEL
      PSI=PSI0+0.883*EPSI
      CALL SOLN2
      TN(5)=REAL(FN2)
      TD(5)=D2
      DEL=DEL0-0.883*EDEL
      CALL SOLN2
      TN(6)=REAL(FN2)
      TD(6)=D2
      DEL=DEL0+0.883*EDEL
      PSI=PSI0-0.883*EPSI
      CALL SOLN2
      TN(7)=REAL(FN2)
      TD(7)=D2
      DEL=DEL0-0.883*EDEL
      CALL SOLN2
      TN(8)=REAL(FN2)
      TD(8)=D2
      FN2MIN=AMIN1(TN(1),TN(2),TN(3),TN(4),TN(5),TN(6),TN(7),TN(8))
      FN2MAX=AMAX1(TN(1),TN(2),TN(3),TN(4),TN(5),TN(6),TN(7),TN(8))
      D2MIN=AMIN1(TD(1),TD(2),TD(3),TD(4),TD(5),TD(6),TD(7),TD(8))
      D2MAX=AMAX1(TD(1),TD(2),TD(3),TD(4),TD(5),TD(6),TD(7),TD(8))
      WRITE(6,495) D2MIN,D2MAX,FN2MIN,FN2MAX
  495 FORMAT(39H0CONFIDENCE LIMITS OF FILM THICKNESS ISF8.1,3H TOF8.1/
     1        69H CONFIDENCE LIMITS OF THE REAL PART OF INDEX OF REFRACTI
     2ON OF FILM ISF12.5,3H TO F12.5)
      GO TO (345,347,348,670),K
  670 FN2X(1)=FN2X(1)+DFN2(1)
      IF(FN2X(1).LT.FN2E(1))  GO TO 343
  700 FN2X(2)=FN2X(2)+DFN2(2)
      IF(FN2X(2).LT.FN2E(2))  GO TO 342
      FNSX(2)=FNSX(2)+DFNS (2)
      IF  (FNSX(2).LT.FNSE(2))  GO TO 327
      FNSX(1)=FNSX(1)+DFNS(1)
      IF(FNSX(1).LT.FNSE(1))    GO TO 326
      WL=WL+DWL
      IF(WL.LT.WLE)             GO TO 320
  705 AI=AI+DAI
      IF(AI.LT.AIE)             GO TO 312
      FN1=FN1+DFN1
      IF(FN1.LT.FN1E)           GO TO 311
      GO TO 20
C        CALCULATE REFRACTIVE INDEX OF SUBSTRATE CNS
  710 FNS=FN1*S1/C1*CSQRT(1.-4.*RHO*S1**2/(RHO+1.)**2)
      FNSI(1)=REAL(FNS)
      FNSI(2)=-AIMAG(FNS)/FNSI(1)
      FNSE(1)=0.
```

```
      FNSE(2)=0.
      WRITE(6,720) FNSI,DEL,PSI
  720 FORMAT(20H0INDEX OF SUBSTRATE=F9.4,F9.4,2H  ,2X,4HDEL=,
     1 F9.3,2X,4HPSI=F9.3)
      IF(FNSI(2).LT.0.) GO TO 725
      GO TO 705
  725 WRITE(6,727)
  727 FORMAT(62H0KAPPA IS SET EQUAL TO 0,SINCE A NEGATIVE VALUE WAS CALC
     1ULATED)
      FNSI(2)=0.
      GO TO 705
C       COMPUTE CONSTANTS OF WAVE PLATE        CWP
C     ASSUMES ANGLE OF TILT AT IS ZERO
  730 Z1=Z1+ALP
      Z2=Z2+ALA
      Z3=Z3+ALP
      Z4=Z4+ALA
      Z5=Z5+ALQ
      WRITE(6,27) L,Z1,Z2,Z3,Z4,Z5
      IF(ABS(Z3-Z5).LT.0.1 .OR. ABS(Z1-Z5).LT.0.1) GO TO 745
  735 B=(TAND(Z2)*(TAND(Z3-Z5)*TAND(Z5)-TAND(Z1-Z5)/TAND(Z5))-TAND(Z4)*
     1  (TAND(Z1-Z5)*TAND(Z5)-TAND(Z3-Z5)/TAND(Z5)))/2./
     2  (TAND(Z2)-TAND(Z4))
      DELC= ATAN2(SQRT(-B*B-TAND(Z1-Z5)*TAND(Z3-Z5)),B)*57.29578
      TC=1./SQRT(-TAND(Z1-Z5)*TAND(Z3-Z5))
      RHOC=TC*CEXP((0.,-1.)*DELC/57.29578)
      RHO =TAND(Z2)*(TAND(Z5)+TAND(Z1-Z5)*RHOC)/(TAND(Z1-Z5)*TAND(Z5)*
     1      RHOC-1.)
      DEL=57.29578*ATAN2(AIMAG(RHO),REAL(RHO))
      PSI=ATAND(CABS(RHO))
      WRITE(6,740) TC,DELC,DEL,PSI
  740 FORMAT(4H TC=,F10.5,9H DELTA C=,F10.5,7H DELTA=,F15.5,5H PSI=,
     1     F15.5)
      GO TO 20
  745 WRITE (6,747)
  747 FORMAT( 73H0CONSTANTS OF WAVE PLATE CAN NOT BE ACCURATELY CALCULAT
     1ED FROM ABOVE DATA)
      GO TO 20
C           CALCULATE ANGLE OF TILT OF STAGE ON ELLIPSOMETER CAT IS COMP
C           ANGLE WITH REAL PART ANGLE AND IMAGINARY PART ERROR
  750 Z1=Z1+ALP
      Z2=Z2+ALA
      Z3=Z3+ALP
      Z4=Z4+ALA
      Z5=Z5+ALQ
      WRITE(6,27 ) L,Z1,Z2,Z3,Z4,Z5
      TP1=TAND(Z1-Z5)
      TA1=TAND(Z2)
      TP2=TAND(Z3-Z5)
      TA2=TAND(Z4)
      TQ =TAND(Z5)
      ALP1=TA1*(TQ+TP1*RHOC)
      ALP2=TA2*(TQ+TP2*RHOC)
      GAM1=TP1*TQ*RHOC-1.
      GAM2=TP2*TQ*RHOC-1.
      BET1=ALP1/SIND(Z2)/COSD(Z2)+(TA1-ALP1*TP1*RHOC/GAM1)/COSD(Z5)**2
      BET2=ALP2/SIND(Z4)/COSD(Z4)+(TA2-ALP2*TP2*RHOC/GAM2)/COSD(Z5)**2
      CAT=(ALP2*GAM1-ALP1*GAM2)/(BET1*GAM2-BET2*GAM1)*57.29578
      ATX=REAL(CAT)
      Z1=Z1+ATX
      Z2=Z2+ATX
      Z3=Z3+ATX
```

```
      Z4=Z4+ATX
      Z5=Z5+ATX
C          COMPUTE RHO,DEL,PSI
      GO TO 174
  755 PRINT 760,CAT,DEL,PSI
  760 FORMAT(15H ANGLE OF TILT=,F13.5,15H DEGREES,ERROR=,F13.5,
     1       7H DELTA=,F15.5,5H PSI=,F15.5)
      IF(M.EQ.26) ATX=(ATX+AT)/2.
      AT=ATX
      GO TO 20
  810 STOP
C  INSTRUCTION CNK  M=31  COMPUTES N AND KAPPA OF FILM
  850 D2=D2I
  852 PRINT 855,D2
  855 FORMAT(1H0,90X,F15.1)
  860 CALL MIN2V(FN2I(1),DFN2(1),FN2E(1),FN2I(2),DFN2(2),FN2E(2))
      D2=D2+DD2
      IF(D2.LT.D2E) GO TO 852
      GO TO 20
      END
@FOR, IS SOLN2
      SUBROUTINE  SOLN2
C         SOLVES FOR REAL PART OF REFRACTIVE INDEX AND THICKNESS OF
C         A FILM FOR MEASUREMENTS OF DELTA AND PSI
C         ER2 IS ERROR OF SOLUTION,THICK IS FILM THICKNESS,INDEX IN FN2
C         IF K=1 OR 3, FIRST(POSITIVE SIGN) SOLUTION IS USED
C         IF K=2 OR 4, SECOND(NEGATIVE SIGN) SOLUTION IS USED
      COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
     1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
     2 T4,D2,DELX ,PSIX , ELDEL,ELPSI,DENP,FNP,DEN1,M,K,IS
      COMPLEX FNX,FN2,CRP,CRN,CRP32,CRN32,RP21,RN21,C2,C3,T1,T2,T3,T4,
     1  T5,T6,RHO,RC
C        S IS USED FOR CALCULATING SPECIFIC REFRACTION
      S(XN)=(XN*XN-1.)/(XN*XN+2.)
      ATAND(X)=ATAN(X)*57.29578
      X1=FN2X(1)
      X3=FN2X(1)+DFN2(1)
      GO TO (10,20,10,20),K
   10 ER1=AIMAG(T1)
      ER3=AIMAG(T3)
      GO TO 30
   20 ER1=AIMAG(T2)
      ER3=AIMAG(T4)
   30 DO 80 I=1,20
        X2=(X1+X3)/2.
        FN2=CMPLX(X2,-FN2X(2)*X2)
        CALL CRC
        CALL CALDX(T5,T6)
      GO TO (40,50,40,50),K
   40   D2=REAL(T5)
        ER2=AIMAG(T5)
        GO TO 60
   50   D2=REAL(T6)
        ER2=AIMAG(T6)
C         COMPUTE DEL AND PSI FOR FILM THICKNESS D2
   60   RHO=RC(RP21,CRP32,FN2*C2*D2/WL)/RC(RN21,CRN32,FN2*C2*D2/WL)
      DELX =ATAN2(AIMAG(RHO),REAL(RHO))*57.29578-DEL
      PSIX=ATAND(CABS(RHO))-PSI
       IF(ABS(DELX ).LT.ELDEL.AND. ABS(PSIX ).LT.ELPSI)  GO TO 100
C         HAS NOT CONVERGED TO ELDEL AND ELPSI.  CHANGE LIMITS X1,X3
C               IF ER1 AND ER2 HAVE SAME SIGN,X1=X2 AND ER1=ER2,OTHERWISE
C                  ER2 AND ER3 HAVE SAME SIGN,X3=X2 AND ER3=ER2
```

```
      IF((ER1.GT.0.0 .AND. ER2.GT.0.0).OR.(ER1.LT.0.0 .AND.
    1 ER2.LT.0.0)) GO TO 70
      X3=X2
      ER3=ER2
      GO TO 80
   70 X1=X2
      ER1=ER2
   80 CONTINUE
   90 D2=1.0E30
      FN2=1.0E30
      RETURN
C     IF FILM THICKNESS D2 IS NEGATIVE,TREAT AS NO SOLUTION
  100 IF(D2.LT.0.) GO TO 90
C       PRINT SOLUTION AND ERROR LIMITS
      IF(M.EQ.17 .OR. M.EQ.22)  GO TO 120
      DIFN21=X2-FN1
      DIFN32=REAL(FNX(N+1))-X2
      FRACTN=DENP/(DENP+(S(X2)-S(FNP))*DEN1/(S(FN1)-S(X2)))
      WTMT=1.0E-5*D2*FRACTN/(FRACTN/DENP+(1.-FRACTN)/DEN1)
      FN2R=REAL(FN2)
      FNXR=REAL(FNX(N+1))
      FK2=-AIMAG(FN2)/REAL(FN2)
      FKX=-AIMAG(FNX(N+1))/REAL(FNX(N+1))
      PRINT 110,D2,FN2R,FK2,WTMT,DELX,PSIX,ER2,DIFN21,DIFN32,
     1 FRACTN,FNXR,FKX,K
  110 FORMAT(1H ,F8.1,2F9.5,1PE12.4,0P2F9.3,F12.5,3F10.5,2F9.5,I6)
      IS=2
  120 RETURN
      END
```

@FOR, IS CRC1, CRC1

```
      SUBROUTINE  CRC
C       CALCULATES CRP32 AND CRN32   REFLECTION COEF OF MEDIUM 2 AND 3
      COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
     1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
     2 T4,D2,DELDX,PSIDX, ERDEL,ERPSI,DENP,FNP,DEN1
      COMPLEX FNX,FN2,CRP,CRN,CRP32,CRN32,RP21,RN21,C2,C3,T1,T2,T3,T4
      COMPLEX RC,RP32,RN32
      SRC(A,B)=(A-B)/(A+B)
      X=FN1*S1
      C2=CSQRT(1.-(X/FN2)**2)
      C3=CSQRT(1.-(X/FNX(3))**2)
      RP21=SRC(FN2*C1,FN1*C2)
      RN21=SRC(FN1*C1,FN2*C2)
      IF(N.EQ.2) GO TO 10
      RP32=SRC(FNX(3)*C2,FN2*C3)
      RN32=SRC(FN2*C2,FNX(3)*C3)
      CRP32=RC(RP32,CRP,FNX(3)*C3*D(3)/WL)
      CRN32=RC(RN32,CRN,FNX(3)*C3*D(3)/WL)
      RETURN
   10 CRP32=SRC(FNX(3)*C2,FN2*C3)
      CRN32=SRC(FN2*C2,FNX(3)*C3)
      RETURN
      END
```

@FOR, IS RCX, RCX

```
      COMPLEX FUNCTION RC(R21,CR32,Q   )
C         RC=REFLECTION COEFFICIENT INCLUDING EFFECT OF UNDERLYING LAYERS
C FOLLOWING CARD IS REQUIRED FOR FORTRAN V ONLY
      COMPLEX CEXP,CLOG,CSQRT
      COMPLEX R21,CR32,Q   ,T
      T=CEXP((0.,-12.56637)*Q)
C         Q=N2*COS(PHI2)*D/WL       (0.,-12.56637)=-4*PI*J
      RC=(R21+CR32*T)/(1.+R21*CR32*T)
```

```
      RETURN
      END
@FOR, IS CALD, CALD
      SUBROUTINE  CALDX(D2C1,D2C2)
C  FOLLOWING CARD IS REQUIRED FOR FORTRAN V ONLY
      COMPLEX CEXP,CLOG,CSQRT
C       CALCULATES FILM THICKNESS AND ERROR TERMS
      COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
     1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
     2 T4,D2,DELDX,PSIDX, ERDEL,ERPSI,DENP,FNP,DEN1,M,K
      COMPLEX FNX,FN2,CRP,CRN,CRP32,CRN32,RP21,RN21,C2,C3,T1,T2,T3,T4
      COMPLEX E,B,C,Y1,Y2,D2C1,D2C2,RHO,T5,T6
      RHO=SIN(PSI/57.29578)/COS(PSI/57.29578)*CEXP((0.,1.)*DEL/57.29578)
C         Y=EXP(-2 J DEL2)    LET  E*Y**2+BY+C=0  Y1,Y2 ARE TWO SOLUTIONS
      E=RHO*RP21*CRP32*CRN32-RN21*CRP32*CRN32
      B=RHO*(CRN32+RP21*RN21*CRP32)-CRP32-RP21*RN21*CRN32
      C=RHO*RN21-RP21
      T5=CSQRT(B**2-4.*E*C)
      GO TO (20,20,10,10),K
C   MAKE IMAGINARY PART OF SQUARE ROOT POSITIVE
   10 IF(AIMAG(T5).LT.0.) T5=-T5
   20 Y1=(-B+T5)/2./E
      Y2=(-B-T5)/2./E
      T5=CLOG(Y1)
      T6=CLOG(Y2)
      D2C1=WL*T5/(0.,-12.56637)/FN2/C2
      D2C2=WL*T6/(0.,-12.56637)/FN2/C2
      RETURN
      END
@FOR, IS MIN2, MIN2
      SUBROUTINE MIN2V(XI,DX,XF,YI,DY,YF)
C        USES FUNCTION FUNC(X,Y) WHICH MUST BE INCLUDED. DETERMINES AND PRINTS
C      VALUES OF X BETWEEN XI AND XF AND VALUES OF Y BETWEEN YI AND YF THAT
C     MINIMIZE FUNC. WILL FIND MULTIPLE MINIMUMS IF THEY ARE SEPARATED
C         BY MORE THAN DX AND DY
      DIMENSION F(50,50)
      COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
     1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
     2 T4,D2,DELDX,PSIDX, ERDEL,ERPSI,DENP,FNP,DEN1,MX,K,IS,RHO
      COMPLEX FNX,FN2,CRP,CRN,CRP32,CRN32,RP21,RN21,C2,C3,T1,T2,T3,T4
      COMPLEX RC,RHO
C     COMPUTE NUMBER OF VALUES OF X AND Y FOR RANGES XI-DX TO XF+DX
C     AND YI-DY TO YF+DY.
      NX=(1.000001*XF-XI)/DX+3
      NY=(1.000001*YF-YI)/DY+3
      IF(NX.GT.50) PRINT 10,DX
   10 FORMAT('0INCREMENT OF X,'E12.5,' IS TOO SMALL')
      IF(NY.GT.50) PRINT 20,DY
   20 FORMAT('0INCREMENT OF Y,'E12.5,' IS TOO SMALL')
C      COMPUTE INITIAL VALUE OF X
      X=XI-DX
C      COMPUTE TABLE F OF FUNCTION FUNC
      DO 40 I=1,NX
         Y=YI-DY
         DO 30 J=1,NY
            F(I,J)=FUNC(X,Y)
C*******DIAGNOSTIC PRINTING*******************************************,
C           PRINT 35,I,J,F(I,J)
   30       Y=Y+DY
   35       FORMAT(2I5,E20.7)
   40    X=X+DX
      NX1=NX-1
```

```
      NY1=NY-1
      DO 910 I=2,NX1
      DO 900 J=2,NY1
C     SEARCH FOR MINIMUMS OF F(VALUES LESS THAN ALL  8 SURROUNDING VALUES)
      IF(F(I,J).LT.F(I-1,J-1).AND. F(I,J).LT.F(I-1,J) .AND.
     1 F(I,J).LT.F(I-1,J+1) .AND. F(I,J).LT.F(I,J-1) .AND.
     2 F(I,J).LT.F(I,J+1) .AND. F(I,J).LT.F(I+1,J-1)
     3 .AND. F(I,J).LT.F(I+1,J) .AND.
     4 F(I,J).LT.F(I+1,J+1))GO TO 50
      GO TO 900
   50 XMIN=XI+(I-2)*DX
      YMIN=YI+(J-2)*DY
      FMIN=F(I,J)
C     A MINIMUM FMIN OCCURS NEAR POINT I,J AT XMIN-DX TO XMIN+DX AND
C       YMIN-DY TO YMIN+DY,
C     SUBDIVIDE RANGE TO LOCATE MINIMUM.
      XD=DX/2
      YD=DY/2
C     SUBDIVIDE RANGE 10 TIMES
      DO 550 K=1,10
      F13=FUNC(XMIN-XD,YMIN-YD)
      F14=FUNC(XMIN,YMIN-YD)
      IF(F13.LT.FMIN) GO TO 110
      IF(F14.LT.FMIN) GO TO 120
C     F14 GT FMIN
  105 F15=FUNC(XMIN+XD,YMIN-YD)
      IF(F15.LT.FMIN) GO TO 130
C     F15 GT FMIN
  107 F18=FUNC(XMIN+XD,YMIN)
      IF(F18.LT.FMIN) GO TO 140
      F22=FUNC(XMIN+XD,YMIN+YD)
      IF(F22.LT.FMIN) GO TO 150
C     F22 GT FMIN
      F21=FUNC(XMIN,YMIN+YD)
      IF(F21.LT.FMIN) GO TO 160
C     F21 GT FMIN
      F20=FUNC(XMIN-XD,YMIN+YD)
      IF(F20.LT.FMIN) GO TO 170
      F17=FUNC(XMIN-XD,YMIN)
      IF(F17.LT.FMIN) GO TO 180
C     FMIN LT SURROUNDINGS
      GO TO 500
  110 IF(F14.LT.F13) GO TO 120
C     F13 LT F14 AND FMIN
      F17=FUNC(XMIN-XD,YMIN)
C     IF F17 LT F13, CONTINUE SCAN WITH F15
      IF(F17.LT.F13) GO TO 105
C     F13 LT F14,FMIN,F17
      F12=FUNC(XMIN-2*XD,YMIN-YD)
      IF(F12.LT.F13) GO TO 190
      F10=FUNC(XMIN-XD,YMIN-2*YD)
      IF(F10.LT.F13) GO TO 200
C     F13 IS NEW MINIMUM
      FMIN=F13
      XMIN=XMIN-XD
      YMIN=YMIN-YD
      GO TO 500
C     F14 LT FMIN,F13
  120 F15=FUNC(XMIN+XD,YMIN-YD)
      IF(F15.LT.F14) GO TO 130
      F11=FUNC(XMIN+XD,YMIN-2*YD)
      IF(F11.LT.F14)  GO TO 210
```

```
      F10=FUNC(XMIN-XD,YMIN-2*YD)
      IF(F10.LT.F14)  GO TO 200
      F17=FUNC(XMIN-XD,YMIN)
      IF(F17.LT.F14) GO TO 107
      F18=FUNC(XMIN+XD,YMIN)
      IF(F18.LT.F14) GO TO 140
C        F14 IS NEW MINIMUM
      FMIN=F14
      YMIN=YMIN-YD
      GO TO 500
C     F15 LT FMIN ,F14
  130 F11=FUNC(XMIN+XD,YMIN-2*YD)
      IF(F11.LT.F15) GO TO 210
      F18=FUNC(XMIN+XD,YMIN)
      IF(F18.LT.F15) GO TO 140
      F16=FUNC(XMIN+2*XD,YMIN-YD)
      IF(F16.LT.F15) GO TO 220
C     F15 IS NEW MINIMUM
      FMIN=F15
      XMIN=XMIN+XD
      YMIN=YMIN-YD
      GO TO 500
C        F18 LT FMIN,F15,F14
  140 F16=FUNC(XMIN+2*XD,YMIN-YD)
      IF(F16.LT.F18) GO TO 220
      F22=FUNC(XMIN+XD,YMIN+YD)
      IF(F22.LT.F18) GO TO 150
      F21=FUNC(XMIN,YMIN+YD)
      IF(F21.LT.F18) GO TO 160
      F23=FUNC(XMIN+2*XD,YMIN+YD)
      IF(F23.LT.F18) GO TO 230
C       F18 IS NEW MINIMUM
      FMIN=F18
      XMIN=XMIN+XD
      GO TO 500
C     F22 LT FMIN,F18
  150 F21=FUNC(XMIN,YMIN+YD)
      IF(F21.LT.F22) GO TO 160
      F23=FUNC(XMIN+2*XD,YMIN+YD)
      IF(F23.LT.F22) GO TO 230
      F25=FUNC(XMIN+XD,YMIN+2*XD)
      IF(F25.LT.F22) GO TO 240
C     F22 IS NEW MINIMUM
      FMIN=F22
      XMIN=XMIN+XD
      YMIN=YMIN+YD
      GO TO 500
C     F21 LT FMIN,F22,F18
  160 F25=FUNC(XMIN+XD,YMIN+2*XD)
      IF(F25.LT.F21) GO TO 240
      F20=FUNC(XMIN-XD,YMIN+YD)
      IF(F20.LT.F21) GO TO 170
      F17=FUNC(XMIN-XD,YMIN)
      IF(F17.LT.F21) GO TO 180
      F24=FUNC(XMIN-XD,YMIN+2*YD)
      IF(F24.LT.F21) GO TO 250
C       F21 IS NEW MINUMUM
      FMIN=F21
      YMIN=YMIN+YD
      GO TO 500
C     F20 LT FMIN,F21
  170 F17=FUNC(XMIN-XD,YMIN)
```

```
      IF(F17.LT.F20) GO TO 180
      F24=FUNC(XMIN-XD,YMIN+2*YD)
      IF(F24.LT.F20) GO TO 250
      F19=FUNC(XMIN-2*XD,YMIN+YD)
      IF(F19.LT.F20) GO TO 260
C     F20 IS NEW MINUMUM
      FMIN=F20
      XMIN=XMIN-XD
      YMIN=YMIN+YD
      GO TO 500
C      F17 LT FMIN F20,F21,F13,F14
  180 F12=FUNC(XMIN-2*XD,YMIN-YD)
      IF(F12.LT.F17) GO TO 190
      F19=FUNC(XMIN-2*XD,YMIN+YD)
      IF(F19.LT.F17) GO TO 260
C     F17 IS NEW MINUMUM
      FMIN=F17
      XMIN=XMIN-XD
      GO TO 500
C     F12 LT F13,FMIN,F17
  190 F10=FUNC(XMIN-XD,YMIN-2*YD)
      IF(F10.LT.F12) GO TO 200
      FMIN=F12
      XMIN=XMIN-2*XD
      YMIN=YMIN-YD
  192 FX=FUNC(XMIN-XD,YMIN-YD)
      IF(FMIN.GT.FX) GO TO 194
      FX=FUNC(XMIN-XD,YMIN)
      IF(FMIN.GT.FX) GO TO 196
      FX=FUNC(XMIN-XD,YMIN+YD)
      IF(FMIN.GT.FX) GO TO 198
      GO TO 500
  194 YMIN=YMIN-YD
  196 XMIN=XMIN-XD
      FMIN=FX
      GO TO 300
  198 YMIN=YMIN+YD
      GO TO 196
C    F10 LT F13,F14,F12,FMIN
  200 FMIN=F10
C   CHECK TO SEE THAT F10 IS NEW MINIMUM
      XMIN=XMIN-XD
      YMIN=YMIN-2*YD
  202 FX=FUNC(XMIN-XD,YMIN-YD)
      IF(FMIN.GT.FX) GO TO 204
      FX=FUNC(XMIN,YMIN-YD)
      IF(FMIN.GT.FX) GO TO 206
      FX=FUNC(XMIN+XD,YMIN-YD)
      IF(FMIN.GT.FX) GO TO 208
      GO TO 500
  204 XMIN=XMIN-XD
  206 YMIN=YMIN-YD
      FMIN=FX
      GO TO 300
  208 XMIN=XMIN+XD
      GO TO 206
C    F11 LT F14,F15
  210 F18=FUNC(XMIN+XD,YMIN)
      IF(F18.LT.F11) GO TO 140
      F16=FUNC(XMIN+2*XD,YMIN-YD)
      IF(F16.LT.F11) GO TO  220
      FMIN=F11
```

```
          XMIN=XMIN+XD
          YMIN=YMIN-2*YD
          GO TO 202
C       F16 LT F15,F11,F18
  220     FMIN=F16
          XMIN=XMIN+2*XD
          YMIN=YMIN-YD
  222     FX=FUNC(XMIN+XD,YMIN-YD)
          IF(FMIN.GT.FX) GO TO 224
          FX=FUNC(XMIN+XD,YMIN)
          IF(FMIN.GT.FX) GO TO 226
          FX=FUNC(XMIN+XD,YMIN+YD)
          IF(FMIN.GT.FX) GO TO 228
          GO TO 500
  224     YMIN=YMIN-YD
  226     XMIN=XMIN+XD
          FMIN=FX
          GO TO 300
  228     YMIN=YMIN+YD
          GO TO 226
C         F23 LT F18,F22,FMIN
  230     F25=FUNC(XMIN+XD,YMIN+2*XD)
          IF(F25.LT.F23) GO TO  240
          FMIN=F23
          XMIN=XMIN+2*XD
          YMIN=YMIN+YD
          GO TO 222
C     F25 LT F21,F22,F23,FMIN
  240     FMIN=F25
          XMIN=XMIN+XD
          YMIN=YMIN+2*YD GO
  242     FX=FUNC(XMIN-XD,YMIN+YD)
          IF(FMIN.GT.FX) GO TO 244
          FX=FUNC(XMIN,YMIN+YD)
          IF(FMIN.GT.FX) GO TO 226
          FX=FUNC(XMIN+XD,YMIN+YD)
          IF(FMIN.GT.FX) GO TO 248
          GO TO 500
  244     XMIN=XMIN-XD
  246     YMIN=YMIN+YD
          FMIN=FX
          GO TO 300
  248     XMIN=XMIN+XD
          GO TO 246
C       F24 LT F21,F20,F17,FMIN
  250     F19=FUNC(XMIN-2*XD,YMIN+YD)
          IF(F19.LT.F24) GO TO 260
          FMIN=F24
          XMIN=XMIN-XD
          YMIN=YMIN+2*YD
          GO TO 242
C      F19 LT F20,F17,FMIN
  260     F17=FUNC(XMIN-XD,YMIN)
          IF(F17.LT.F19) GO TO 180
          FMIN=F19
          XMIN=XMIN-2*XD
          YMIN=YMIN+YD
          GO TO 192
C                  FOLLOW MINIMUM FOR 5 STEPS
  300     DO 400 KK=1,5
              FX=FUNC(XMIN-XD,YMIN-YD)
              IF(FX.LT.FMIN) GO TO 310
```

```
            FX=FUNC(XMIN-XD,YMIN)
            IF(FX.LT.FMIN) GO TO 320
            FX=FUNC(XMIN-XD,YMIN+YD)
            IF(FX.LT.FMIN) GO TO 330
            FX=FUNC(XMIN,YMIN-YD)
            IF(FX.LT.FMIN) GO TO 340
            FX=FUNC(XMIN,YMIN+YD)
            IF(FX.LT.FMIN) GO TO 350
            FX=FUNC(XMIN+XD,YMIN-YD)
            IF(FX.LT.FMIN) GO TO 360
            FX=FUNC(XMIN+XD,YMIN)
            IF(FX.LT.FMIN) GO TO 370
            FX=FUNC(XMIN+XD,YMIN+XD)
            IF(FX.LT.FMIN) GO TO 380
            GO TO 500
  310       YMIN=YMIN-YD
  320       XMIN=XMIN-XD
            GO TO 400
  330       YMIN=YMIN+YD
           GO TO 320
  340       YMIN=YMIN-YD
            GO TO 400
  350       YMIN=YMIN+YD
            GO TO 400
  360       YMIN=YMIN-YD
  370       XMIN=XMIN+XD
            GO TO 400
  380       YMIN=YMIN+YD
            GO TO 370
  400       FMIN=FX
C                A MINIMUM IS NOT FOUND IN 5 STEPS
         GO TO 600
  500    XD=XD/2
         YD=YD/2
  550    CONTINUE
C                      NEW MINIMUM IS FOUND
      GO TO 700
  560 FORMAT(5E15.6,3I5/(7F15.6))
  600 PRINT 610,K,FMIN,XMIN,YMIN,XD,YD
  610 FORMAT('0 ITERATION NUMBER'I3,' FAILED FOR MINIMUM'E12.5,' AT X
     1=,E12.5,' Y='E12.5,' RANGES =' 2E12.5)
C****************PRINT VALUES AT MINIMUM
  700 FN2=CMPLX(XMIN,-XMIN*YMIN)
      T1=RC(RP21,CRP32,FN2*C2*D2/WL)/RC(RN21,CRN32,FN2*C2*D2/WL)
      DDEL=ATAN2(AIMAG(T1),REAL(T1))*57.29578-DEL
      DPSI=ATAN(ABS(T1))*57.29578-PSI
    XD=2*XD
    YD=2*YD
      PRINT 800, XMIN,YMIN,XD,YD,FMIN,DDEL,DPSI
  800 FORMAT(' '5F15.5,2F10.3)
  900 CONTINUE
  910 CONTINUE
    RETURN
    END
```

@FOR, IS FUNC

```
    FUNCTION FUNC(FRN,FK)
C     COMPUTES VALUES OF DELTA AND PSI, DELC AND PSIC, FOR A FILM
C        OF N=FRN,KAPPA=FK, AND THICKNESS=D2 AND ERROR TERM,FUNC
    COMMON FN1,S1,C1,N,FNX(900),D(900),FN2,CRP,CRN,CRP32,CRN32,WL,
   1 RP21,RN21,DEL,PSI,C2,C3,FN2I(2),DFN2(2),FN2E(2),FN2X(2),T1,T2,T3,
   2 T4,D2,DELDX,PSIDX, ERDEL,ERPSI,DENP,FNP,DEN1,MX,K,IS,RHO
    COMPLEX FNX,FN2,CRP,CRN,CRP32,CRN32,RP21,RN21,C2,C3,T1,T2,T3,T4
```

```
COMPLEX RC,RHO
   FN2=CMPLX(FRN,-FRN*FK)
   CALL CRC
FUNC=ABS(RC(RP21,CRP32,FN2*C2*D2/WL)/RC(RN21,CRN32,FN2*C2*D2/WL)
1  -RHO)
RETURN
END
```

```
@PRT, T TPF $.
@MAP, IL A,B
LIB SYS*MSFC $.
@XQT, B
BLANK CARD
DATA CARDS
@FIN
```

APPENDIX B

P AND A VALUES AS FUNCTIONS OF INCIDENCE FOR LIF/AL AND AL MIRRORS FOR ZONES I AND 2



180°
90°
0°
PARAMETERS
P2
P2
A2
A2
ZONE 2
P1
P1
A1
A1
ZONE 1
Al
LIF/Al
50°
60°
70°
80°
90°
ANGLE OF INCIDENCE

APPENDIX C

MEASURABLE ELLIPSOMETRIC PARAMETERS FOR SEVERAL OPAQUE REFLECTORS

A. P and A Values for Several Types of Mirrors

MIRROR	A	P
Au	54.23 150.90	4.25 95.75
Al	150.25 58.25	74.00 163.25
Lif/Al	144.25 41.63	8.93 99.30
Al/Au	153.10 51.15	8.93 98.70
DC-704 Oil/Au	146.25 55.00	3.50 92.13
Unknown Contaminant/Lif/Al	149.45 51.50	85.85 177.77

B. P and A Values Versus Angle of Incidence and 1/4 λ Plate Settings of Al Mirror

Angle of Incidence, degrees	1/4 λ Plate setting, degrees	A	P
70	-45	53.42 147.25	76.50 163.75
70	+45	57.17 147.50	22.67 111.42
60	-45	56.30 149.20	63.62 153.87
60	+45	50.90 150.75	30.40 121.60
55	-45	57.30 148.70	60.40 150.12
55	+45	49.12 154.45	34.30 124.75

APPENDIX D

CALCULATION OF FILM THICKNESS AS A FUNCTION OF SMALL DELTA FOR A TRANSPARENT FILM ON SILICON

Calculation of film thickness as a function of δ (small delta) for a transparent film on silicon [3] where

$$\delta\,(\text{real}) = \frac{360d}{\lambda} \left[\left(N_1^2 - K_1^2 - \sin^2\Theta\right)^2 + 4N_1^2K_1^2\right]^{\frac{1}{4}} \times \cos\left[\frac{1}{2}\tan^{-1}\left(\frac{-2N_1K_1}{N_1^2 - K_1^2 - \sin^2\Theta}\right)\right] .$$

Index of refraction is 1-10.

Small delta (δ)	Thickness (d), nm
0.0	0.000
0.05	0.01925
0.25	0.09627
0.50	0.1925
2.50	0.9627
5.00	1.9254
7.50	2.888
10.0	3.85
20.0	7.70
30.0	11.50
40.0	15.40
50.0	19.25
60.0	23.10
70.0	26.95
80.0	30.80
90.0	34.658
100.0	38.508
110.0	42.35
120.0	46.21
130.0	50.06
140.0	53.9125
150.0	57.76
160.0	61.6143
170.0	65.46
180.0	69.316

APPENDIX E

THICKNESS OF A TRANSPARENT FILM ON AN OPAQUE SUBSTRATE AS A FUNCTION OF THE MEASURABLE ELLIPSOMETRIC PARAMETERS Δ AND ψ


260
240
220
200
180
160
140
120
100
0
10
20
30
40
50
Δ(DELTA)
Ψ(PSI)
140
130
150
120
160
110
170
100
180
90
80
10
70
20
60
30
50
40

REFERENCES

1. Faucher, J. A.; McManus, G. M.; and Trurnit, H. J.: Simplified Treatment of Ellipsometry. J. O. S. A., vol. 48, 1958, pp. 51-54.

2. McCrackin, Frank L.: A Fortran Program for Analysis of Ellipsometer Measurements. National Bureau of Standards, Technical Note 479, April 1969.

3. Archer, R. J.: Manuel on Ellipsometry. Gaertner Scientific Corporation, 1968.

4. Instruction Manual 7109-C-116-RO for Ellipsometers L118 and L119 Series. Gaertner Scientific Corporation.

5. Feynman, R. P.; Leighton, R. B.; and Sands, M.: The Feynman Lectures on Physics. Addison Wesley Company, Inc., New York, Vol. I, II, 1965.

6. Hurlbut, Cornelius S.: Dana's Manual of Mineralogy. John Wiley Sons, Inc., New York, 1971.

7. Young, Hugh D.: Fundamentals of Optics and Modern Physics. McGraw-Hill Book Company, New York, 1971.

APPROVAL

THE GAERTNER L119 ELLIPSOMETER AND ITS USE IN THE MEASUREMENT OF THIN FILMS

By Michael Linkous

The information in this report has been reviewed for security classification. Review of any information concerning Department of Defense or Atomic Energy Commission programs has been made by the MSFC Security Classification Officer. This report, in its entirety, has been determined to be unclassified.

This document has also been reviewed and approved for technical accuracy.

William C. Snoddy

WILLIAM C. SNODDY
Chief, Space Thermophysics Division

W. Haeussermann

W. HAEUSSERMANN
Acting Director, Space Sciences Laboratory

DISTRIBUTION

INTERNAL

AD-S
Dr. E. Stuhlinger

A& PS-MS-IP(2)

PD-AP-DIR
Mr. J. Downey

PD-MP-DIR
Mr. H. Gierow

SL/SE-ATM
Mr. E. Cagle

SL-EI
Mr. C. David

S& E-ASTR-RP
Dr. J. Randall

S& E-SSL-DIR
Dr. W. Haeussermann
Mr. R. Hembree

S& E-SSL-X
Mr. H. Weathers

S& E-SSL-N
Dr. E. Urban

S& E-SSL-P
Dr. R. Naumann
Mr. W. Moore

S& E-SSL-S
Dr. W. Sieber

S& E-SSL-T
Mr. W. Snoddy
Dr. K. Schocken
Mr. E. Miller
Mr. B. Jones
Mr. G. Arnett
Mr. R. Linton
Mr. J. Zwiener
Dr. R. Kroes
Mr. S. Fields
Mr. J. Reynolds
Mr. M. Linkous (10)

S& E-SSL-C
Reserve (5)

A &PS-PAT
Mr. L. D. Wofford, Jr.

A &PS-MS-H
A &PS-MS-IL (8)
A &PS-TU (6)

EXTERNAL

Scientific and Technical Information
Facility (25)
P. O. Box 33
College Park, Maryland 20740
Attn: NASA Representative (S-AK/RKT)